#被动技能【双重打击】
execute as @a[tag=akeer,scores={Diamond_sword1=1..}] run function nw:yingxiong/3yingxiong/03akeer/skill/0shuangchongdaji
execute as @a[tag=akeer,scores={skIron_sword1=1..}] run function nw:yingxiong/3yingxiong/03akeer/skill/0fanhui

#主动技能【反复横跳】触发：左平移
execute as @a[tag=akeer,scores={skEgg=1}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/1fanfuhengtiao0
#主动技能【反复横跳】二段：右平移
execute as @a[tag=akeer,scores={skCDPlayer1=1..,skEgg=3}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/1fanfu1
#主动技能【反复横跳】三段：左平移
execute as @a[tag=akeer,scores={skCDPlayer1=1..,skEgg=5}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/1fanfu2
#主动技能【反复横跳】四段：右平移
execute as @a[tag=akeer,scores={skCDPlayer1=1..,skEgg=7}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/1fanfu3
#主动技能【反复横跳】五段：4s后换回主武器
execute as @a[tag=akeer,scores={skCDPlayer1=0}] run clear @s iron_sword
execute as @a[tag=akeer,scores={skCDPlayer1=0}] run clear @s diamond_sword
execute as @a[tag=akeer,scores={skCDPlayer1=0}] run give @s minecraft:diamond_sword{Damage:1558,AttributeModifiers:[{Operation: 0, Amount: +4.5d, UUID: [I; 506346098, -762989019, -376979115, 1505498882], Slot: mainhand, AttributeName: "generic.attack_damage", Name: "generic.attackDamage"},{Operation: 0, Amount: -2.4d, UUID: [I; 503545885, -156791109, -457453409, -1924971166], Slot: mainhand, AttributeName: "generic.attack_speed", Name: "generic.attackSpeed"}],display:{Name:'[{"text":"纳提克斯之剑III","color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]',Lore:['{"text":"被动技能【双重打击】","color":"dark_purple","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}','{"text":"akeer的每3次攻击将进行快速的双重打击","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}','{"text":"#获得一把0耐久铁剑【纳提克斯之剑】","color":"red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}','{"text":"（+10攻击速度、+2.5攻击伤害）","color":"red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}','{"text":"#而后重新回到【纳提克斯之剑III】","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}']}} 1
execute as @a[tag=akeer,scores={skCDPlayer1=0,skEgg=1..}] run effect clear @s invisibility
execute as @a[tag=akeer,scores={skCDPlayer1=0,skEgg=1..}] run scoreboard players set @s skCDEgg 18
execute as @a[tag=akeer,scores={skCDPlayer1=0}] run scoreboard players reset @s skCDPlayer1

#主动技能【反复横跳】被动每次攻击-1
execute as @a[tag=akeer,scores={Diamond_sword0=1..}] run scoreboard players remove @s[scores={skCDEgg=0..}] skCDEgg 1
scoreboard players reset @a[tag=akeer,scores={Diamond_sword0=1..}] Diamond_sword0
execute as @a[tag=akeer,scores={skIron_sword0=1..}] run scoreboard players remove @s[scores={skCDEgg=0..}] skCDEgg 1
scoreboard players reset @a[tag=akeer,scores={skIron_sword0=1..}] skIron_sword0
#主动技能【反复横跳】CD
execute as @a[tag=akeer,scores={skCDEgg=..0}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/1fanfuhengtiao_cd
#主动技能【野性呼唤】特效
execute as @a[tag=akeer,tag=mingxiang] at @s rotated ~ 0 positioned ^ ^ ^-0.1 run particle composter ~ ~ ~ 0.3 1 0.3 0.01 5
execute as @a[tag=akeer,tag=mingxiang] at @s rotated ~ 0 positioned ^ ^ ^-0.1 run particle minecraft:totem_of_undying ~ ~2 ~ 0.3 0.1 0.3 0.01 2
#主动技能【野性呼唤】效果
effect give @a[tag=akeer,tag=mingxiang] resistance 1 1 true
effect give @a[tag=akeer,tag=mingxiang,scores={skCDEntity=1..}] regeneration 2 2 false
scoreboard players reset @a[tag=akeer,scores={skCDEntity=1..}] skCDEntity

#主动技能【野性呼唤】行走打断效果
execute as @a[tag=akeer,tag=mingxiang,scores={skWalkDistance=35..}] run function nw:yingxiong/3yingxiong/03akeer/skill/2huhuanreset
scoreboard players reset @a[tag=akeer,scores={skWalkDistance=35..}] skWalkDistance
#主动技能【野性呼唤】疾跑打断效果
execute as @a[tag=akeer,tag=mingxiang,scores={skRunDistance=60..}] run function nw:yingxiong/3yingxiong/03akeer/skill/2huhuanreset
scoreboard players reset @a[tag=akeer,tag=mingxiang,scores={skRunDistance=60..}] skRunDistance
#主动技能【野性呼唤】攻击打断效果
execute as @a[tag=akeer,tag=mingxiang,scores={skShangHai=1..}] run function nw:yingxiong/3yingxiong/03akeer/skill/2huhuanreset
scoreboard players reset @a[tag=akeer] skShangHai
#主动技能【野性呼唤】跳跃打断效果
execute as @a[tag=akeer,tag=mingxiang,scores={skJump=1..}] run function nw:yingxiong/3yingxiong/03akeer/skill/2huhuanreset
scoreboard players reset @a[tag=akeer] skJump

#主动技能【野性呼唤】
execute as @a[tag=akeer,scores={skSnowball=1..}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/2huhuan
#主动技能【野性呼唤】CD
execute as @a[tag=akeer,scores={skCDSnowball=..0}] run function nw:yingxiong/3yingxiong/03akeer/skill/2huhuan_cd

#主动技能【丛林的狩猎者】
execute as @a[tag=akeer,scores={skEXBottle=1}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/3dazhao
#主动技能【丛林的狩猎者】特效
execute as @a[tag=akeer,scores={akeer5=1..}] at @s rotated ~ 0 positioned ^ ^ ^-0.1 run particle composter ~ ~ ~ 0.3 1 0.3 0.01 3
execute as @a[tag=akeer,scores={akeer5=1..}] at @s rotated ~ 0 positioned ^ ^ ^-0.1 run particle minecraft:totem_of_undying ~ ~ ~ 0.3 1 0.3 0.01 5
#主动技能【丛林的狩猎者】效果
effect clear @a[tag=akeer,scores={akeer5=1..}] slowness
effect give @a[tag=akeer,scores={akeer5=1..}] haste 1 2 true
effect give @a[tag=akeer,scores={akeer5=1..}] speed 1 1 true
#主动技能【丛林的狩猎者】增加技能持续时间
scoreboard players add @a[tag=akeer,scores={akeer5=1..,skKill=1..}] akeer5 5
scoreboard players remove @a[tag=akeer,scores={skKill=1..,skCDEXBottle=1..}] skCDEXBottle 4
scoreboard players remove @a[tag=akeer,scores={skKill=1..,skCDSnowball=1..}] skCDSnowball 4
scoreboard players remove @a[tag=akeer,scores={skKill=1..,skCDEgg=1..}] skCDEgg 4
scoreboard players reset @a[tag=akeer,scores={skKill=1..}] skKill
#主动技能【丛林的狩猎者】CD
execute as @a[tag=akeer,scores={skCDEXBottle=..0}] at @s run function nw:yingxiong/3yingxiong/03akeer/skill/3dazhao_cd

#C03阿克尔（丛林猎手）未开启狩猎者时
execute as @a[tag=akeer,scores={akeer5=0}] run title @s actionbar [{"text":"反复横跳","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDEgg","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 野性呼唤","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDSnowball","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 丛林的狩猎者","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDEXBottle","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]

#C03阿克尔（丛林猎手）开启狩猎者后
execute as @a[tag=akeer,scores={akeer5=1..}] run title @s actionbar [{"text":"反复横跳","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDEgg","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 野性呼唤","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"正在冷却：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDSnowball","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 丛林的狩猎者","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 剩余时间：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"akeer5","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]
